
CONTENIDOS

En esta unidad vamos a recoger una serie de aplicaciones didácticas de la Hoja de Cálculo, 
dirigidas  a  diferentes  niveles  y  disciplinas  y  aprovechando  diferentes  propiedades  del 
programa.

Si bien vamos a mezclar en cada caso funciones de diferente tipo de la Hoja de Cálculo, 
hemos organizado los ejemplos en las siguientes categorías:

● La hoja de cálculo como organizador de la información

● Funciones de texto

● Funciones de Hoja de Cálculo

● Registro,  análisis y manejo de datos.

Por  supuesto  los  ejemplos,  centrado  cada  uno  en  un  contexto  determinado,  pueden 
aplicarse  a  otras  muchas  situaciones.  Lo  que  se  pretende  es  mostrar  un  catálogo  de 
herramientas  y  trucos  para  que  cada  uno  pueda  diseñar  actividades  adaptadas  a  su 
realidad educativa.

LA HOJA DE CÁLCULO COMO ORGANIZADOR DE LA INFORMACIÓN

Las primeras aplicaciones didácticas que vamos a introducir se basan en las posibilidades 
de la hoja de cálculo para organizar la información. Por su distribución en celdas, es un 
programa idóneo cuando queremos presentar datos de forma estructurada. Además, la 
posibilidad  de  dar  un  formato  diferente  a  cada celda  permite  muchas  alternativas  de 
presentación.

Ejemplo 1: Sopa de letras

En primer lugar, vamos a elaborar una sopa de letras. Para ello, abrimos un nuevo archivo 
de Calc.  Nos situamos en primer lugar en la  Hoja 2 y seleccionamos las 10 primeras 
columnas y situando el ratón entre dos de ellas, aparecerá la doble flecha que nos permite 
modificar el  ancho. Al  pinchar aparecerá el  ancho en cm y lo reducimos a un tamaño 
menor de 1 cm. A continuación seleccionamos las diez primeras filas y botón derecho> 
Altura de fila  y fijamos la misma cantidad.

Ya hemos conseguido que las celdas tengan forma cuadrada. Ahora elaboramos la sopa de 
letras siguiendo los siguientes pasos:

1. Escribimos las palabras solución en horizontal, vertical o diagonal.

2. Sin soltar la tecla Ctrl, seleccionamos con el ratón todas las palabras solución.

3. Con  botón derecho >Formatear celdas> Efectos de fuente cambiamos las 
palabras a rojo.

4. Rellenamos el resto de celdas con otras letras cualesquiera.
5. Centramos  las  letras  en  las  celdas  seleccionando  todas  volviendo  al  cuadro 

Formatear celdas: clic derecho>Formatear celdas>Alineación> Alineación de 
texto y allí  elegimos  Centrado tanto en horizontal como en vertical.  Con estos 
pasos ya tenemos la solución.

Para reproducir el pasatiempo sin solución en la Hoja 1, basta escribir en cada celda que 
su valor sea el mismo que el de la Hoja 2. (p.e. en A1 escribimos la fórmula =Hoja2.A1). 
Lo interesante es que se reproduce el texto sin formato por lo que desaparece el rojo de 
las soluciones
Volvemos a centrar las letras y para que tenga más aspecto de pasatiempo quitamos las 
líneas de división con  menú Herramientas> Opciones> OpenOffice.org Calc>Ver y 
desmarcamos Líneas de cuadrícula. 
Por último añadimos un borde y escribimos unas indicaciones para que el usuario sepa lo 
que tiene que buscar en la sopa.

Si quieres dar un aspecto más profesional repasa algunos de los puntos tratados en la 
sesión 2 Formato>Contenidos>Interpretación y seguridad.

Ahora  vamos  a  desarrollar  una  estrategia  para  que  el  propio  programa  compruebe 
automáticamente el pasatiempo. De momento tenemos en la Hoja1 la sopa de letras con 
las indicaciones oportunas y el formato adecuado; en la Hoja2 están las soluciones que 
podremos ocultar cuando sea necesario.
En la Hoja3 vamos a activar una casilla que nos permita acceder a la solución. Para ello 

(después de copiar de nuevo la sopa, igual que hicimos en la Hoja1), seleccionamos todas 
las letras pertenecientes a las palabras escondidas y las aplicamos el siguiente Formato 
condicional:

● Primero  definimos  un  nuevo  estilo  en  Estilo  y  formato llamado  Rojo cuya 
característica es que la fuente sea de color rojo y en negrita.

● Después asignamos el  formato condicionado con la condición  de que si  en una 
cierta  casilla  (en  este  caso  L1)  escribimos  una  S (de  “solución”)  las  celdas 
seleccionadas aparezcan en Rojo.

Puedes probar este modelo accediendo al archivo sopa.ods

Ejemplo 2: Crucigrama

El otro ejemplo que vamos a presentar que aprovecha la estructura de la Hoja de Cálculo 
es la elaboración de un crucigrama.
En  primer  lugar,  preparamos  la  solución  dejando  algunas  filas  en  blanco  (podemos 
empezar en la fila 25). Para ello:

1. Cambiamos ancho y alto de las celdas (p.e. 0,82 cm) para formar celdas cuadradas.

2. Escribimos las palabras solución poniendo un número al principio de cada palabra.

3. Seleccionando palabra a palabra le ponemos borde de cuadrícula

4. Alineamos las letras en el centro, los números pegados a su palabra.

Para preparar el pasatiempo:

5. Copiamos el crucigrama y lo pegamos en la parte de arriba de la hoja

6. Volvemos a preparar las celdas cuadradas.

7. Pulsamos la tecla Supr y desmarcamos la casilla Formatos.

Ya tenemos el crucigrama mudo. Para que se compruebe de forma automática, vamos a 
echar mano del Formato Condicional.

Primero con  menú Formato> Estilo y formato,  definimos uno  Nuevo que se llame 
“bien” y cuyo Estilo de celda sea que el Fondo sea de color verde.
Una vez definido,  elegimos cualquiera  de las  celdas  y en  menú Formato> Formato 
condicional definimos  uno  que  si  el  valor  de  la  celda  es  igual  al  valor  de  la 
correspondiente de la solución el estilo sea “bien”.

Nota. Es importante poner E23 en lugar de $E$23 para que la referencia sea relativa y 
podamos copiar el formato en otras celdas. Ahora bien, eso implica que si trasladamos la 
solución a otras celdas, el formato condicionado fallará.

En la celda que hemos formateado, elegimos Copiar, vamos seleccionando cada palabra 
(hay  que  hacerlo  de  una  en  una)  y  botón  derecho> Pegado  especial y  dejamos 
marcada sólo la casilla Formatos.

De esta manera, al ir rellenando el crucigrama, las letras correctas aparecerán en fondo 
verde mientras que las demás aparecerán con fondo blanco.
El crucigrama se completa escribiendo las definiciones, como puedes ver en el archivo 
crucigrama.ods

FUNCIONES DE TEXTO

Hay diversas funciones de la Hoja de Cálculo que son herramientas de trabajo con los 
textos  incluidos  en  las  celdas,  de  manera  que  podemos  comparar,  trocear,  unir... 
diferentes textos.

Para ilustrar esto abre el  modelo  letras.ods.  Este modelo permite simular un conocido 
juego en el que dadas dos palabras de igual número de letras (en el modelo vamos a 
utilizar palabras de cuatro letras) hay que ir convirtiendo la primera palabra en la segunda 
cambiando cada vez una sola de sus letras.

Para  ver  cómo  hemos  conseguido  que  el  modelo  funcione  accede  a  menú 
Herramientas>  Proteger  documento> Hoja.  Una  vez  desprotegido,  selecciona  las 
columnas C y M, haz clic derecho (sobre la cabecera de dichas columnas)> Mostrar de 
forma que se visualicen las columnas D a L.

La estructura de la hoja es la siguiente (puedes comprobarlo jugando una “partida” con las 
columnas desprotegidas):

1. En la columna C las celdas están desprotegidas de manera que el jugador pueda ir 
escribiendo las sucesivas palabras.

● En las  columnas  D a  G extraemos  cada  una  de  las  letras  de la  palabra 
mediante precisamente fórmulas del tipo =EXTRAE($C7;2;1) (extrae de la 
palabra que hay en C7, a partir de la letra en segundo lugar, una letra)

● En las columnas  H a  K comparamos cada una de las cuatro letras  de la 
palabra  con  las  correspondientes  de  la  anterior.  Para  ello  utilizamos  la 
función =IGUAL() que  compara  dos  textos  y  devuelve  Verdadero  o  Falso 
según el caso. Para que aparezca un 1 ó un 0 añadimos la función =VALOR( ) 
que  convierte  el  texto  en  número.  La  función  es,  pues,  del  tipo 
=VALOR(IGUAL(G12;G11)). En la columna  L sumamos los cuatro valores 
(lo que equivale al número de letras cambiadas)

● En  M introducimos una fórmula de control para evitar trampas. En ella se 
incluyen los siguientes elementos:

1. Una función =ESBLANCO( ) para que no aparezca ningún mensaje si la 
celda de la palabra está vacía.

2. Si la suma de los contadores es 3, es que se han cambiado tres letras 
y mostramos el mensaje “Vale”. Si es 4, la palabra es la misma que la 
anterior y escribimos “No has cambiado ninguna”.

3. Para el resto de casos (0,1 ó 2) elaboramos un mensaje, mediante la 
función  =CONCATENAR(),  que  combine  el  texto  “No  vale,  has 
cambiado” con el número que resulta de restar a 4 el resultado de la 
suma.

=CONCATENAR("No vale, has cambiado ";4-L7)

La  función  CONCATENAR(  )  es  especialmente  útil  pues  nos  permite  combinar  textos 
entrecomillados con fórmulas.

Sesión 6– Aplicaciones didácticas 7 de 17

http://modelos6/letras.ods


● En N situamos la fórmula que nos anuncia de forma automática que se ha 
conseguido el objetivo. Para ello incluye:

1. Una  función  =SI()  que  compara  cada  palabra  con  la  solución. 
Anidada en esta hay otra función =SI()  que comprueba además 
que el número de cambios ha sido 3 para que no salte el mensaje 
de aprobación si hemos puesto la palabra correcta haciendo más 
de un cambio en el último paso. Para que aparezca el número de 
intentos restamos a 40 (número máximo de intentos que permite 
la  hoja)  el  número  de  celdas  en  blanco,  que  contamos  con  la 
función  =CONTAR.BLANCO($C$7:$C$46)

FUNCIONES DE HOJA DE CÁLCULO

Aquí nos referiremos a funciones que de alguna forma se refieren al propio funcionamiento 
del programa. En primer lugar vamos a introducir la función =BUSCARV()

Ejemplo 1: la función BUSCARV

Como aplicación presentamos un pequeño traductor  de nombres de colores.  Para realizarlo, 
escribimos en la columna B los nombres en castellano de una serie de colores y en la columna C 
su traducción al inglés.

La función BUSCARV nos va permitir la traducción simultánea de un idioma al otro. Para 
ello escribimos en la celda A1 el nombre de uno de los colores de la lista, por ejemplo el 
VERDE. En D1 escribimos la siguiente fórmula:

=BUSCARV(A1;B1:C24;2;0)

Donde:

1. A1 contiene la palabra que queremos buscar

2. B1:C24 es el rango de celdas donde vamos a buscarla

3. Una vez localizada la fila de esa palabra (en este caso VERDE), el 2 nos indica 
el contenido de qué celda de esa fila queremos que se muestre (de las dos de la 
fila del VERDE, la segunda, es decir, GREEN)

4. El  0 indica  que  la  columna  donde  buscamos  no  está  ordenada  de  forma 
ascendente. Si los datos que utilizamos sí lo están, escribimos 1 u omitimos el 
valor.

Con lo que, en este caso, en D1 aparecerá GREEN.

Para que no se vea la tabla, seleccionamos las dos columnas B y C y Botón secundario> 
Ocultar

Podemos realizar las siguientes mejoras, como puedes ver en el modelo buscarv.ods:

● Podemos hacer que sea el alumno el que escriba la solución y comprobar si  es 
correcta. Para ello ocultamos también la columna D y preparamos una celda donde 
se  escriba  la  traducción  y,  mediante  una  función  =SI(  ) la  cotejamos  con  la 
solución, presentando mensajes del tipo: “Muy bien” / “Prueba otra vez”.

● Si la solución es incorrecta, podemos intentar que la Hoja de Cálculo muestre la 
verdadera solución en la frase usando la función Concatenar. La fórmula sería del 
tipo  =CONCATENAR(“La solución es”;J15),  siendo  J15  la  celda  que  contiene 
dicha solución.

Nota. Para que funcione esta última solución varias veces, hay que borrar primero cada 
vez  el  valor  de  la  segunda  celda  para  evitar  que  se  interprete  como  una  solución 
incorrecta del nuevo intento.

Ejemplo 2: Manejo de errores

Abre  el  modelo  romanos.ods.  En  él  se  incluye  un  sencillo  modelo  que  permite  la 
conversión de números romanos en nuestro sistema decimal y viceversa.

Para ese fin OpenOffice Calc dispone de dos fórmulas sencillas: =ROMANO() que convierte 
a números romanos y =ÁRABE() (¡cuidado, con acento!) que convierte a nuestro sistema 
decimal de origen árabe.

Con estas dos funciones elaborar el modelo de conversión es trivial. Pero nos queremos 
centrar en el manejo de los errores. Para entender a qué nos referimos desprotege la hoja 
y selecciona las columnas D y G y botón secundario> Mostrar que precisamente nos 
mostrará las columnas E y F que habíamos ocultado.

Hemos establecido unos códigos para elegir el tipo de conversión deseada. Elige 1 para 
pasar de “árabes” a romanos. En la columna E se incluye la fórmula  =ROMANO(D7) que 
no ofrece más complicación. Pero, si escribimos un número árabe demasiado grande, el 
resultado será de error.

De  la  misma  manera,  si  eliges  el  código  2,  y  escribes  números  romanos  la  función 
=ÁRABE(D9) funciona  perfectamente  para  cualquier  número  romano  pero,  como  es 
natural,  da  lugar  a  error  cuando  escribimos  cualquier  letra  que  no  corresponde  con 
números romanos o combinaciones incorrectas de las que sí corresponden.

¿Cómo  evitamos  esto?  Calc incluye  una  función  que  es  =TIPO.DE.ERROR(),  que 
devuelve el número de error existente en una celda determinada. Si has provocado errores 

en los dos tipos de conversión verás que en las celdas de la columna F aparece el error 
número 502,  correspondiente  al  error  “Argumento  no  válido”  que  es  lo  que sucede si 
elegimos un número demasiado grande o una combinación inexistente de letras.

Para que aparezca en blanco cuando hay un error, procedemos como sigue:

● Definimos un estilo en menú Formato>Estilo y formato> Nuevo que se llame 
Error y le incluimos como única característica que en  Efecto de fuente el color 
sea blanco.

● Seleccionamos  las  celdas  G7  y G9  y  le  aplicamos  el  siguiente  Formato 
condicional:

Así si el resultado en F7 (análogamente para F9) es 502, el formato aparece con Estilo Error (con 
color de fuente blanco).

Ejemplo 3: Funciones a la vista

Abre ahora el modelo corrector.ods. Este modelo está pensado para “poner ejercicios” de 
forma  rápida  y  sencilla.  En  la  primera  columna  de  la  Hoja  2  escribimos  operaciones 
matemáticas  con  la  complejidad  que  se  desee.  Una  vez  escritas,  la  Hoja  calcula  las 
soluciones y podemos ocultarlas seleccionado la columna y botón secundario> Ocultar 
e incluso se puede Proteger después la Hoja.

Después  en  la  Hoja1,  que  es  la  que  será  presentada  al  alumnado,  escribimos  en  la 
columna  A fórmulas del tipo (por ejemplo en la celda A2)  =FORMULA(Hoja2.A2). Esta 
función de Calc muestra la fórmula que se ha incluido en otra celda, con lo que esta toma 
forma de enunciado de ejercicio. 

Dejamos la columna B para que se escriban las soluciones y en la C, mediante fórmulas 
del tipo  =IGUAL(B2;Hoja2.A2) comparamos las respuestas con las soluciones de la Hoja 
2. (Hemos añadido una función =SI(ESBLANCO()) para que sólo actúe cuando exista la 
respuesta). La función IGUAL devuelve mensajes Verdadero/Falso.

REGISTRO Y ANÁLISIS DE DATOS

En este apartado trataremos de las posibilidades que tiene la Hoja de Cálculo para sacar 
partido a un conjunto de datos que obtengamos.

Nos servirá de base para la explicación un ejemplo de Educación Física que utiliza los 
registros obtenidos por una clase en cinco pruebas físicas. Normalmente en las clases de 
la materia se suele trabajar con un mayor número de pruebas, pero para simplificar el 
ejemplo nos vamos a limitar a cinco:  Abdominales,  Flexibilidad,  Lanzamiento de balón, 
Cooper (carrera continua de 12´) y 60 metros. Para estas pruebas existen unos baremos 
establecidos que puntúan de 0 a 10 según la edad y género de cada participante.  El 
ejemplo está pensado para una clase de 4º de la E.S.O. Los baremos para este curso 
quedan recogidos en la Hoja 4ESO del archivo marcas.ods.

Nota. Para ver mejor el contenido de las hojas es recomendable establecer un 75% o así 
en menú Ver> Escala.

El objetivo del modelo con el que vamos a trabajar es recoger las marcas de todos los 
alumnos de una clase dos veces en un mismo curso, de forma que podamos medir su 
progresión de una a otra medición. El modelo nos permitirá puntuar cada registro de forma 
que podamos determinar la nota global, elaborar un informe global de la clase así como 
informes individuales de tantos alumnos como queramos.

Abre,  si  no  lo  has  hecho  ya,  el  modelo  marcas.ods.  Como  puedes  comprobar,  está 
compuesto  de  cuatro  hojas:  la  última,  como  hemos  dicho,  recoge  los  baremos 
correspondientes a este nivel y a estas pruebas. La primera,  resultados, es la hoja de 
registro  de todos los  datos.  La hoja  infor glob recoge un informe general  del  curso, 
mientras que infor indiv contiene un informe del alumno que elijamos.

Veamos el proceso de elaboración de cada una:

Registro de datos

Las primeras celdas de la primera hoja permiten rellenar los datos relativos al grupo con el 
que se trabaja, el curso durante el que se han tomado los datos así como el nombre del 
profesor/a, datos todos ellos que se rescatarán en otros puntos del archivo.

La hoja  de registro en sí  se  basa en conjuntos  de celdas  como el  que aparece en la 
imagen. Los datos se recogen en las celdas de fondo blanco. En las grises se calcula la 
puntuación mientras que en las de fondo violeta aparece la progresión.

Para calcular la puntuación echamos mano de una fórmula como la siguiente:

=SI(C4="M";BUSCARV(D4; 4ESO.$A$3:$C$63; 2; 0);SI(C4="F";BUSCARV(D4; 
4ESO.$A$3:$C$63; 3; 0); " "))

Donde utilizamos la función BUSCARV() para buscar el registro en el rango A3:C63 de la 

hoja 4ESO, que es donde se encuentran las puntuaciones de la prueba de abdominales. 
Con las funciones  SI() distinguimos si  el  género es masculino o femenino para que la 
función  BUSCARV()  devuelva  los  datos  de  las  columnas  segunda  o  tercera, 
respectivamente.

Para medir el progreso, la fórmula utilizada es:

=SI(ESBLANCO(D4); " "; SI(E4=0;" "; (G4-E4)/E4))

El cálculo es en realidad (G4-E4)/E4, que divide la diferencia de las mediciones entre la 
primera de ellas. Al dotar a la celda de formato porcentaje expresa el % de mejora (o, si es 
negativo, de pérdida) de una a otra medición. 

Para evitar problemas, añadimos unos condicionantes en caso de que no existan datos, 
con la función  ESBLANCO(),  o que la primera medición sea 0, con lo que no se puede 
dividir  entre esa cantidad (en ese caso,  de hecho,  podríamos decir  que si  la  segunda 
marca ya no es cero, el aumento en porcentaje sería de alguna forma infinito, lo cuál por 
cierto haría muy feliz al alumno sobre todo si lo que ha sacado es un 1)

Lo mismo hacemos por supuesto con las otras cuatro pruebas. Para facilitar el registro de 
las marcas, sería cómodo fijar la parte correspondientes a los datos del alumnado. Eso lo 
conseguimos sin más que seleccionar las columnas C y D (entre las cuales estaría el límite 
de las  dos partes)  y  menú Ventana> Fijar.  Para anularlo  bastaría  con desmarcar  la 
opción.

La hoja de registro de datos se completa con unas últimas columnas en las que se calcula 
el total de puntuación en cada evaluación entre las cinco pruebas y la media de las cinco 
que es la nota final. Igual que en cada prueba, en el total se calcula el progreso habido. 

También  se  incluye  una  última fila,  en  la  que,  mediante  la  función  =PROMEDIO(  ) se 
calculan las marcas y notas medias de las clases en cada prueba y evaluación.

Resumen de resultados

Una vez que se han rellenado todos los datos, la segunda hoja (infor glob) ofrece un 
análisis global del grupo con el que se está trabajando. En él se incluyen los siguientes 
elementos:

1. Los datos de grupo, curso y profesor/a, rescatados de la primera hoja.

2. Los resultados globales del grupo por prueba, los cuáles se obtienen de la 

fila  de  promedios  de  la  primera  hoja,  pero  con  una  mejora:  las  notas 
suspensas aparecen en rojo y las aprobadas en verde.

Para conseguir ese efecto, definimos en menú Formato> Eslilo y formato> Nuevo los 
estilos  suspenso  (con fondo rojo suave y color de fuente blanco)  y  aprobado (fondo 
verde suave y color de fuente blanco)

Después  seleccionamos  las  celdas  de  puntuación  y  menú  Formato>  Formato 
condicional, asignándole uno u otro formato según la nota sea menor que 5 o mayor o 
igual que 5. 

3. Un gráfico estadístico que recoge las puntuaciones por prueba comparando 
las dos evaluaciones. Para elaborarlo seleccionamos con el ratón solamente 
la columna con los nombres de las pruebas y las dos correspondientes a 
puntuación. (para lo que nos ayudamos de la tecla Ctrl )

Como las puntuaciones oscilan entre 0 y 10 obligamos en las Propiedades del eje Y que la 
escala oscile entre esos valores.

4. Una tabla que analiza los resultados por género.

● En  la  última  tabla  de  la  hoja  recogemos  resultados  globales  pero 
diferenciados por género. Para ello vamos a utilizar dos funciones que 
nos permiten operar sólo para algunos valores cada vez: CONTAR.SI y 
SUMAR.SI

● En primer lugar, para contar el número de alumnas y el de alumnos, 
usamos fórmulas del tipo:

=CONTAR.SI(resultados.$C$4:$C$31; "F")

● Que realiza un recuento en la columna C (correspondiente al género) 
sólo  de  aquellas  celdas  que  incluyen  “F”  (femenino).  Lo  mismo 
hacemos con “M” para recontar el número de alumnos.

Para calcular la nota media de, por ejemplo, las alumnas, debemos sumar solamente las 
notas medias de las alumnas y dividir esta suma entre el número de alumnos. Después 
hacemos lo mismo con los alumnos. Lo conseguimos con esta fórmula:

=(SUMAR.SI(resultados.$C$4:$C$31; "F"; resultados.$AD$4:$AD$31))/B$36
              

(1)    (2)  (3)     (4)

La  función   =SUMAR.SI  suma  sólo  aquellas  celdas  que  cumplen  una  determinada 
condición, de manera que:

1. Es el rango de celdas donde comprobamos la condición (la columna de género)

2. Es la condición (en este caso que aparezca un determinado texto) para que el valor 
se incorpore a la suma.

3. Es el rango de las celdas que sumamos si se cumple la condición (este rango no 
tiene por qué ser el mismo que aquel en el que comprobamos la condición). Es la 
columna de las notas en la evaluación correspondiente (en este caso la primera).

Una vez hecha la suma condicionada la dividimos entre el número de alumnas, (4), para 
calcular la media.

Datos individuales

La última Hoja se dedica a mostrar la información de un alumno que elijamos. Esta hoja 
incluye:

● Número y nombre del alumno/a. Se pide el número de lista y la hoja localiza el 

nombre mediante una función =BUSCARV().

● Un resumen de sus puntuaciones en las pruebas. Todas ellas se consiguen mediante 
fórmulas del tipo:

=BUSCARV(E7;resultados.B4:AB31; 4; 0)
Donde buscamos el nombre en un rango que incluye todos los resultados y sólo hay que 
determinar la columna donde se sitúan los datos de la prueba correspondiente.

Al igual que hicimos en el resumen global, incluimos un Formato Condicionado de manera 
que salgan los suspensos en fondo rojo y los aprobados en verde.

● Un gráfico de barras que recoge los resultados anteriores. Para que se visualice 
mejor, en la propiedades del Eje X, desmarcamos la opción Mostrar etiqueta y por 
el contrario, en las propiedades de una de las series de barras, marcamos dicha 
opción. Además, como hicimos en el gráfico anterior obligamos a que la escala sea 
de 0 a 10.

● Una última tabla que calcule la posición de los resultados del alumno/a respecto del 
conjunto de la clase.  Con este fin,  utilizamos una función =BUSCARV() que nos 
busque la puntuación de cada evaluación. 

Para saber su posición en la clase introducimos la siguiente fórmula:

=JERARQUÍA(F35;resultados.AC4:AC31)
Que devuelve el lugar que ocupa un valor (en este caso la puntuación del alumno en una 
de las evaluaciones) dentro de un rango (la columna de las puntuaciones de todo el grupo 
para esa evaluación.

Este  es  el  contenido  del  modelo.  Las  dos  hojas  resumen,  global  e  individual,  están 
diseñadas para ocupar  un folio,  lo  que facilita su impresión.  Se puede comprobar con 
menú Archivo> Vista preliminar.

COMPLEMENTOS

Números aleatorios.

La  Hoja  de  Cálculo  Calc  genera  números  aleatorios (al  azar)  mediante  la  función 
=ALEATORIO( ),  la cuál devuelve un número decimal al  azar entre 0 y 1. Además con 
menú Herramientas> Contenidos de celdas> Recalcular (o mucho más rápido con 
su atajo, la tecla F9) recalcula y muestra otro número al azar entre 0 y 1.

Esto nos permite simular lanzamientos de monedas o dados, con las siguientes fórmulas:

=ENTERO(ALEATORIO()*2) Devuelve sólo los valores 0 y 1 (moneda)

=1+ENTERO(ALEATORIO()*6) Devuelve valores del 1 al 6 (dado)

Hay una función que nos permite elegir un valor aleatorio, entre 1 y 6 por ejemplo, que es 
la función =ALEATORIO.ENTRE(1;6) pero es una función que no se recalcula al pulsar F9, 
es más sofisticado pero se recalcula manteniendo pulsadas a la vez las teclas Mayúscula 
y Ctrl. y sin soltar pulsar F9 las veces que se estime oportuno.

El  modelo  caballos.ods incluye  una  aplicación  de  los  números  aleatorios.  Se  simulan 
tiradas de dos dados en una carrera de caballos numerados del 2 al 12. Cuando los dos 
dados suman el número del caballo, este avanza una posición y el proceso se repite hasta 
que uno de ellos llega a meta.
Esta simulación permite comprobar a los alumnos que, al tirar dos dados, no todas las 
sumas  tienen la  misma probabilidad  (hay  dos  o  tres  “caballos”  claramente  favoritos). 
Además se cuenta el número de intentos con lo que, una vez determinados los caballos 
ganadores, se puede estudiar el tiempo que necesitan para la ”victoria”, etc...

PRÁCTICAS.

Práctica 1. Crucigrama o sopa

Diseña  un  crucigrama  o  sopa  de  letras  utilizando  alguno  de  los  conceptos  de  tu 
especialidad o nivel educativo y que tenga algún tipo de corrección automática.

Práctica 2. Siglo

Elabora un archivo de hoja de cálculo que, dado un año antes o después de Cristo, nos  
diga a qué siglo pertenece. 
Para ello echa mano de la función ROMANO( ), teniendo en cuenta que los años pueden 
tener signo positivo o negativo, por lo que habrá que actuar de forma diferente en cada 
caso. También debes tener cuidado porque si  el  año 2001 pertenece al S.XXI,  el  2000 
pertenece al XX.
Para escribir expresiones del tipo “S.II a.de C. puedes utilizar la función CONCATENAR( ).

Práctica 3. Radianes

Realiza  un modelo que convierta  la  medida de un ángulo  en grados a  su medida en  
radianes. 
Un radián corresponde a un ángulo cuyo arco tiene la misma longitud que el radio, por lo 
que los 360º de una circunferencia equivalen a 2π radianes ya que la longitud de una 
circunferencia es 2π por el radio.

Para realizar el modelo, debes usar la función =RADIANES() que, precisamente convierte el 
número de grados en radianes. 
Si escribes, por ejemplo =RADIANES(360) el resultado será 6,28 (resultado de multiplicar 
por dos la aproximación de π=3,14).

Para que quede más elegante, nuestro objetivo es que el modelo exprese los radianes en 
función de π. Para ello deberás dividir el valor entre π (su valor en la Hoja de Cálculo viene 
dado por la función =PI()) y después hacer que aparezca la letra π (que es la letra p con el 
tipo de fuente Symbol) a continuación del número usando la función =CONCATENAR()

